Newton-féle érintő módszer

Newton-féle érintő módszer

A módszer

Sok esetben az $f(x) = 0$ egyenlet gyökeinek, azaz az f függvény zérushelyeinek "pontos" meghatározása nem lehetséges. Galois (francia matematikus, 1811-1832) megmutatta, hogy ötödfokú, vagy magasabb fokú polinomok zérushelyeinek meghatározására nincs megoldóképlet, tehát olyan eljárás amely a négy alapművelet és a gyökvonás segítségével szimbolikusan megadható, pontos értékeket ad. Nem is beszélve az olyan egyenletekről, mint például az

$$\mathbf{e}^x = 2 - x^2$$

egyenlet!

Általában azonban ilyenkor is lehetőség van a gyökök tetszőleges pontosságú meghatározására.

Első lépésként igyekszünk a lehetséges gyökökről hozzávetőlegesen tájékozódni.

A megoldandó egyenletet $\mathrm{f}(x) = 0$ alakra hozzuk.

Ezután felrajzoljuk a szereplő függvény grafikonját, esetleg több különböző szakaszon.

A fenti esetben az

$$\mathbf{e}^x + x^2 - 2 = 0$$

egyenlet gyökei az

$$\mathrm{f}(x) = \mathbf{e}^x + x^2 - 2$$

függvény zérushelyei. Ábrázoljuk tehát az f függvényt!

```
> restart:
> plot(exp(x)+x^2-2,x=-3..2,color=blue,thickness=2);
```



>

Az ábráról látszik, de könnyen igazolható is, hogy két gyök van, és az egyik a $[-2, -1]$, a másik a $[0, 1]$ intervallumban van. Ebben az esetben gyorsan" túlestünk" azon az eljáráson, amit a gyökök elkülönítésének neveznek. Ennek keretében meghatározzuk azokat az intervallumokat, amelyekben a függvénynek pontosan egy gyöke van. Ezek után a gyököket a Newton- féle érintőmódszerrel akkor határozhatjuk meg, ha a kérdéses $[a, b]$ intervallumon (ez itt a $[-2, -1]$, illetve a $[0, 1]$ intervallum) teljesülnek a következő tételben szabott feltételek.

Tétel

Teljesüljenek az f függvényre az $[a, b]$ intervallumon a következő feltételek.

1. Legyen f az $[a, b]$ intervallumon kétszer deriválható (akkor folytonos is).

2. Ne legyen az f deriváltjának zérushelye, azaz $\frac{d}{dx}\,\mathrm{f}(x) \neq 0$ az [a,b] intervallumon (akkor f szigorúan monoton).

3. Ne legyen a második deriváltnak zérushelye, azaz $\frac{d^2}{dx^2}\,\mathrm{f}(x) \neq 0$ az $[a,b]$ intervallumon (akkor f-nek nincs inflexiós pontja).

Ha mindezek mellett $f(a)\,f(b) < 0$, tehát az f előjele különböző az intervallum végpontjaiban, akkor az $[a,b]$ intervallumon az f függvénynek pontosan egy zérushelye van.

Vizsgáljuk meg, hogy a következő függvény teljesíti-e a tétel feltételeit:

```
> f:=x->2*x^3+x-1;
```

$$f := x \to 2\,x^3 + x - 1$$

Newton-féle érintőmódszer előkészítése

Első dolgunk a tétel feltételeinek megfelelő intervallum kiválasztása. Olyan intervallomot keresünt tehát, ahol sem az első, sem pedig a második derivált nem vesz fel nulla értéket, ugyanakkor az intervallum végpontjaiban a függvényértékek ellenkező előjelűek.

Rajzoljuk fel a függvényt!

```
> plot(f,-2..2);
```



Első látásra a [0,1] intervallum megfelelőnek tűnik. Ellenőrizzük azonban a tétel feltételeit! Nézzük meg az első és a második derivált viselkedését a kiszemelt intervallumon. A feltételek szerint egyik sem lehet nulla az vizsgált intervallumban.

```
> D(f);
```

$$x \to 6\,x^2 + 1$$

```
> (D@@2)(f);
```

$$x \to 12\,x$$

Az első derivált láthatón teljesíti feltételt, hiszen az egész számegyenesen pozitív, és így a [0,1] intervallumon is az. A második derivált azonban x=0 helyen zérus értéket vesz fel, így a 0 nem szerepelhet a kiszemelt intervallumban. Azonban a nullánál kicsivel nagyobb érték már megfelel kívánalmaknak. Választásszuk a [0.3,1] intervallumot

```
> a:=0.3;b:=1;
```

$$a := 0.3$$

$$b := 1$$

Vigyáznunk kellett azonban arra, hogy nehogy túl nagy értéket adjuk a-nak mert a végpontokban a függvényértékeknek ellenkező előjelűeknek kell lennie. Így pl. az $a = 0.9$ választás már bizonyosan hibás lenne. A mi választásunk azonban helyes, mert

```
> signum(f(a)*f(b));
```

$$-1$$

```
>
```

A tétel állítása szerint tehát ebben az intervallumban a függvénynek pontosan egy gyöke van.
Az előkészítés utolsó lépéseként válaszzuk ki az $[a,b]$ intervallumnak azt a végpontját, ahol függvényérték előjele megegyezik a második derivált előjelével és jelöljük ezt a pontot x_0-lal. Esetünkben a második derivált pozitív az $[a,b]$ intervallumon, így a helyes választás $x_0 = b$.

Newton-féle érintőmódszer első lépése

Húzzunk érintőt az f függvény görbéjéhez a $P(x_0, f(x_0))$ pontban. A gyököt az érintő x tengellyel való metszéspontjával közelítjük. Jelöljül ezt a metszéspontot x_1-gyel, az alábbi ábrának megfelelően.

```
> x0:=b:x1:=-f(x0)/D(f)(x0)+x0:
  ticks:=[[x0=`x0=b`,x1=`x1 `],[]]:
  kep:=plot([[x0,0],[x0,f(x0)],[-f(x0)/D(f)(x0)+x0,0]],color
  =red):
  szv:=plots[textplot]([x0-0.1,f(x0),`P0(x0,f(x0))`]):
  rajz:=plot(f(x),x=x0-0.7..x0+0.1,color=blue,thickness=2,ti
  ckmarks=ticks):
  plots[display]([rajz,kep,szv]);
```



A $P_0(x_0, \mathrm{f}(x_0))$ ponton átmenő érintő egyenlete a következő.

```
> y-f(x0)=D(f)(x0)*(x-x0);
```

$$y-2=7\,x-7$$

Az $y=0$ helyettesítést elvégezve, az egyenletet rendezve adódik a gyök első közelítő értéke. Figyeljük meg, hogy x_1 közelebb van a gyökhöz, mint x_0, tehát x_1 a keresett gyök jobb közelítését adja.

```
> x1:=evalf(solve(subs(y=0,%),x));
```

$$x1 := 0.7142857143$$

Newton-féle érintőmódszer második lépése

Az ötlet már a kezünkben van. Most azt tesszük, hogy az első lépésben meghatározottakat megismételjük x_0, helyett x_1-re. Tehát most a $\mathrm{P}(x_1, \mathrm{f}(x_1))$
pontban húzunk érintőt. A keresett gyök újabb közelítését ennek az érintőnek az x tengellyel való metszéspontja adja, melyet x_2-vel jelölünk, az alábbi ábrának megfelelően.

```
> x0:=b:x1:=-f(x0)/D(f)(x0)+x0:x2:=-f(x1)/D(f)(x1)+x1:
  ticks:=[[x0=`x0=b`,x1=`x1 `,x2=`x2 `],[]]:
  kep0:=plot([[x0,0],[x0,f(x0)],[-f(x0)/D(f)(x0)+x0,0]],colo
  r=red):
  szv0:=plots[textplot]([x0-0.1,f(x0),`P0(x0,f(x0)`]):
  kep1:=plot([[x1,0],[x1,f(x1)],[-f(x1)/D(f)(x1)+x1,0]],colo
  r=red):
  szv1:=plots[textplot]([x1-0.1,f(x1),`P1(x1,f(x1)`]):
  rajz:=plot(f(x),x=x0-0.7..x0+0.1,color=blue,thickness=2,ti
  ckmarks=ticks):
  plots[display]([rajz,kep0,szv0,kep1,szv1]);
```



Most felírjuk a $P_1(x_1, \mathrm{f}(x_1))$ ponton átmenő érintő egyenletét, melyből az y=0 helyettesítést elvégezve, az egyenletet rendezve adódik a gyök második közelítő értéke

```
> y-f(x1)=D(f)(x1)*(x-x1);
```

$$y-\frac{152}{343}=\frac{199\,x}{49}-\frac{995}{343}$$

```
> x2:=evalf(solve(subs(y=0,%),x));
```

$$x2 := 0.6051687006$$

Newton-féle érintőmódszer további lépései

Világos, hogy az első és második lépésben megmutatott eljárást minden határon túl folytathatjuk. Így valós számok egy x_0, x_1, x_2, ... végtelen sorozatához jutunk, amelyről megmutatható hogy konvergens, és határértéke a keresett gyök. Mi ennek a végtelen sorozatnak az x_0-ból kiindulva csak két további elemét számoltuk ki.

```
> x0,x1,x2;
```

$$1, \frac{5}{7}, 0.6051687006$$

Azt is látjuk azonban, hogy "végtelen sok" számértéket a gyakorlatban nem tudunk előállítani. De nincs is rá szükség. mert abból, hogy az x_n sorozat a gyökhöz tart, az következik, hogy $f(x_n)$ a nullához tart, így annak ε sugarú környezetéből az $f(x_n)$ sorozatnak csak véges sok eleme marad ki. Más szóval, akármilyen kis ε értéket is veszünk fel, véges sok lépés után eljutunk olyan x_{n_0} értékhez, amelyre $|f(x_{n_0})| < \varepsilon$ Ezt az x_{n_0} értéket a keresett gyök megfelelő közelítésének tekinthetjük. Ezzel a észrevétellel elértük azt, hogy Newton-féle érintőmódszer véges sok lépés után végetér.

Newton-féle érintőmódszer összefoglalása

Legyen az f az $[a,b]$ intervallumon kétszer differenciálható, és tegyük fel, hogy az intervallumon f-nek sem az első, sem a második deriváltja nem vesz fel nulla értéket. Tegyük fel továbbá, hogy $f(a)$ és $f(b)$ különböző előjelű. Ekkor f-nek az $[a, b]$ intervallumon pontosan egy gyöke van, amit a következő eljárással közelítünk meg. Legyen x_0 az intervallum valamelyik végpontja. Felírjuk a $P_0(x_0, f(x_0))$ ponton átmenő érintő egyenletét

$$y - f(x_0) = D(f)(x_0)(x - x_0)\ ,$$

és meghatározzuk ennek y tengellyel való metszéspontjat, vagyis y =0 után megoldjuk a fenti egyenletet. A keletkező pontot x_1-gyel jelöljük.

$$x_1 = x_0 - \frac{f(x_0)}{D(f)(x_0)}$$

Ezután legyen

$$x_2 = x_1 - \frac{f(x_1)}{D(f)(x_1)}\ ,$$

$$x_3 = x_2 - \frac{f(x_2)}{D(f)(x_2)},$$

....

általánosan

$$x_n = x_{n-1} - \frac{f(x_{n-1})}{D(f)(x_{n-1})}\ , \quad (n = 1, 2, ...)\ .$$

Belátható, hogy a $x_0, x_1, x_2, \ldots$ sorozat konvergens és monoton módon az $f(x)$ függvény egyetlen $[a, b]$ -beli ξ zérushelyéhez tart.

Az eljárást többféleképpen is leállíthatjuk. Egyik lehetséges megoldás, hogy az eljárás akkor ér véget, ha az egymást követő értékek különbségére egy $0 < \varepsilon$ hibakorlátnál kisebb

$$|x_{n+1} - x_n| < \varepsilon\ .$$

érték adódik.

A másik módszer, hogy a függvényértékekre teszünk feltételt, és akkor állunk meg ha

$$|f(x_n)| < \varepsilon\ .$$

Megjegyzés:

A Newton-féle érintő módszer $N(x) = x - \frac{f(x)}{\frac{d}{dx} f(x)}$ függvénnyel végzett iteráció végrehajtását jelenti.

Az **iteráció** kifejezés itt a következőt jelenti: Az N iterációs függvény felhasználásával képezzük az N, $N@N$, $(N@N)@N$, . . .$((N@N)@N)@N$, ...sorozatot, tehát a sorozat n-edik eleme az N -ből képezett n-szeres kompozíció:



Az iterációs sorozat tehát

$$x_1 = N(x_0),$$
$$x_2 = N(x_1),$$
$$x_3 = N(x_2),$$
$$.......$$

általánosan:

$$x_n = N(x_{n-1}), \quad n = 1, 2, . . .$$
$$......$$

A kézi számítás modellezése

A módszer mély bevésése érdekében hasznos legalább néhány esetben a részleteket is figyelemmel kísérni, esetleg néhányszor manuálisan is végigszámolni. Ez hozzásegíthet a mentális kép megfelelő kialakításához és megszilárdulásához.Ezt szolgálja a **newton-**eljárás, amely az iterációs lépéssorozat részletes dokumentációja, tehát a manuális megoldás szimulációja. Az **f függvény**re a ***kezdő*** értékből kiindulva elvégzi az iterációt ***lépésszám*** -szor. Az eljárás működésével itt nem foglalkozunk, mert ezek döntően olyan ismeretek taglalását kívánnák, ami inkább elterelné a figyelmet.

```
> restart:
> newton:=proc(f::algebraic,kezdo,lepesszam)
  local i,x,M,n:
  n:=lepesszam:x:='x':
  M:=matrix(lepesszam+1,5,0):
  M[1,1]:='n':M[1,2]:=x['n']:M[1,3]:='f(x[n])':
  M[1,4]:='D(f)(x[n])':
  M[1,5]:=x['n']-'f(x[n])'/'D(f)(x[n])':
  x[1]:=evalf(kezdo):
  for i to n do
    M[i+1,1]:=i: M[i+1,2]:=x[i]:
```

```
M[i+1,3]:=f(x[i]):M[i+1,4]:=D(f)(x[i]):M[i+1,5]:=x[i]-f(x[
i])/D(f)(x[i]):
   x[i+1]:=M[i+1,5]:
od:
evalm(M);
end:
```

Példa

Határozzuk meg a

$$e^x = 2 - x^2$$

egyenlet pozitív gyökét $10^{(-4)}$ *pontossággal!*

Megoldás

Az $f(x) = e^x + x^2 - 2$ függvény pozitív x értékekre szigorúan monoton növekvő, tehát csak egy zérushelye lehet.

```
> f:=x->exp(x)+x^2-2;
```

$$f := x \to e^x + x^2 - 2$$

Mivel a 0 és a 2 helyen vett függvényértékek szorzata negatív, a gyök a $[0, 1]$ intervallumban van

```
> evalf(f(0)*f(2));
```

-9.389056099

```
> plot([f(x),D(f)(x),(D@@2)(f)(x)],x=0..1,color=[blue,red,gr
een],thickness=2,title=`f=kék,D(f)=piros,(D@@2)(f)=zöld`);
>
```



Láthatóan az első és a második derivált is az egész intervallumon pozitív, tehát a feltételek teljesülnek. Kezdőértékként a zérushelynél nagyobb értéket kell választanunk, mert ekkor egyezik meg a második derivált előjele a függvényérték előjelével. Legyen a kezdőérték 1.

```
> newton(f,1,4);
```

$$\begin{bmatrix} n & x_n & \mathrm{f}(x_n) & \mathrm{D}(f)(x_n) & x_n - \dfrac{\mathrm{f}(x_n)}{\mathrm{D}(f)(x_n)} \\ 1 & 1. & 1.718281828 & 4.718281828 & 0.6358246729 \\ 2 & 0.6358246729 & 0.292851974 & 3.160228305 & 0.5431566927 \\ 3 & 0.5431566927 & 0.016451520 & 2.807745712 & 0.5372973587 \\ 4 & 0.5372973587 & 0.000063825 & 2.785970090 & 0.5372744493 \end{bmatrix}$$

Az eljárás eredményeként megjelenítődik a gyök közelítését szolgáló lépéssorozat, úgy, ahogy azt kézi számolás esetén dokumentálnunk kellene.
Láthatóan 4 lépés még kevés, hiszen a harmadik és a negyedik érték különbsége nagyobb a hibakorlátnál. Legyen a lépésszám eggyel nagyobb, 5.

```
> newton(f,1,6);
```

n	x_n	$f(x_n)$	$D(f)(x_n)$	$x_n - \frac{f(x_n)}{D(f)(x_n)}$
1	1.	1.718281828	4.718281828	0.6358246729
2	0.6358246729	0.292851974	3.160228305	0.5431566927
3	0.5431566927	0.016451520	2.807745712	0.5372973587
4	0.5372973587	0.000063825	2.785970090	0.5372744493
5	0.5372744493	0.	2.785885065	0.5372744493
6	0.5372744493	0.	2.785885065	0.5372744493

Így már a kívántnál is nagyobb pontossággal megkaptuk a gyököt, amelynek értéke négy tizedesjegy pontossággal 0.5373.

Eljárások

A newton eljárás

A **newton-**eljárás az iterációs lépéssorozat részletes dokumentációját adja, tehát a manuális megoldás szimulációja. Az **f függvény**re a ***kezdő*** értékből kiindulva elvégzi az iterációt ***lépésszám***-szor.

```
> restart:
> newton:=proc(f::algebraic,kezdo,lepesszam)
  local i,x,M,n:
  n:=lepesszam:x:='x':
  M:=matrix(lepesszam+1,5,0):
  M[1,1]:='n':M[1,2]:=x['n']:M[1,3]:='f(x[n])':
  M[1,4]:='D(f)(x[n])':
  M[1,5]:=x['n']-'f(x[n])'/'D(f)(x[n])':
  x[1]:=evalf(kezdo):
  for i to n do
    M[i+1,1]:=i: M[i+1,2]:=x[i]:
  M[i+1,3]:=f(x[i]):M[i+1,4]:=D(f)(x[i]):M[i+1,5]:=x[i]-f
  (x[i])/D(f)(x[i]):
     x[i+1]:=M[i+1,5]:
  od:
  evalm(M);
  end:
```

Példa

Tekintsük az

$$f: \mathbf{R}\text{+->}\mathbf{R};\ f(x) = \mathbf{e}^{(-x^2)} - \ln(x)$$

függvényt!

1. Mutassuk meg, hogy a függvénynek egyetlen zérushelye van!

2. Határozzuk meg a zérushelyet a Newton módszerrel, 5 tizedesjegy pontossággal!

Megoldás

1.

A függvény a pozitív valós számok halmazán értelmezett. Meg kell mutatnunk először, hogy van zérushelye, majd azt, hogy csak egy van.

```
> f:=x->exp(-x^2)-ln(x);
```

$$f := x \to e^{(-x^2)} - \ln(x)$$

Ábrázoljuk a függvényt, mondjuk a [0,10] intervallumon!

```
> plot(f(x),x=0..10);
```



Sikeresnek mondhatjuk az ábrázolás intervallumának megválasztását, hiszen láthatóan az [1,2] intervallumon van zérushely. Nem mindig sikerülhet már az első lépésben ez! Ilyenkor kísérleteznünk kell! Változtatgatva az intervallumot több lépés után sikerülhet az, ami most első lépésre "ölünkbe hullott".
A függvény szigorúan monoton csökkenőnek látszik. Az ábra alapján azonban nem következtethetünk a függvénynek a nagyobb változóértékek esetén való viselkedésére. A derivált viszont segítségünkre lehet. Tudjuk, ha egy adott intervallumon negatív az első derivált értéke, akkor azon a számközön a függvény szigorúan monoton csökkenő. Képezzük a deriválat!

```
> Diff(f(x),x)=diff(f(x),x);
```

$$\frac{d}{dx}\left(e^{(-x^2)} - \ln(x)\right) = -2\,x\,e^{(-x^2)} - \frac{1}{x}$$

Azonnal látszik, hogy pozitív x-ekre a derivált negatív, tehát f szigorúan monoton csökkenő. Ezzel igazoltuk, hogy a függvénynek nincs több zérushelye.

2. A Newton-algoritmus alkalmazhatóságához már csak a második deriváltat kell megvizsgálnunk. Olyan, a zérushelyet tartalmazó intervallumot kell keresnünk, amelyben a második derivált állandó előjelű.

Ábrázoljuk a második deriváltat a zérushelyet tartalmazó [1, 2] intervallumon!

```
> plot((D@@2)(f)(x),x=1..2,y=0..2);
```




```
>
```

Láthatóan a második derivált az egész intervallumon pozitív, tehát az f függvény az egész számközön alulról konvex. A kezdőértéket úgy kell megválasztanunk, hogy a kiszemelt pontban a függvényérték előjele a második derivált előjelével megegyező legyen. Rajzoljuk föl most együtt a függvényt s az első két deriváltat, s az ábrát is figyelve döntsünk a kezdőértékről!

```
> plot([f(x),D(f)(x),(D@@2)(f)(x)],x=1..2,color=[blue,red
  ,green],thickness=2,legend=["f","D(f)","(D@@2)(f)"]);
```



>

Láthatóan az $x = 1$ megfelelő kezdőérték. Hívjuk meg a newton-eljárást az f függvényre, az $x = 1$ kezdőértékre, és 4 lépésre!

```
> newton(f,1,4);
```

n	x_n	$f(x_n)$	$D(f)(x_n)$	$x_n - \frac{f(x_n)}{D(f)(x_n)}$
1	1.	0.3678794412	-1.735758882	1.211941558
2	1.211941558	0.0379773557	-1.383102667	1.239399647
3	1.239399647	0.0005892417	-1.340320382	1.239839275
4	1.239839275	$0.1484\ 10^{-6}$	-1.339642169	1.239839386

A két utolsó érték különbsége még nagyobb a kívánt pontosságnál, thát emelnünk kell a lépésszámot!

```
> newton(f,1,5);
```

n	x_n	$f(x_n)$	$D(f)(x_n)$	$x_n - \frac{f(x_n)}{D(f)(x_n)}$
1	1.	0.3678794412	-1.735758882	1.211941558
2	1.211941558	0.0379773557	-1.383102667	1.239399647
3	1.239399647	0.0005892417	-1.340320382	1.239839275
4	1.239839275	$0.1484\ 10^{-6}$	-1.339642169	1.239839386
5	1.239839386	$-0.2\ 10^{-9}$	-1.339641998	1.239839386

>

Most már megállapíthatjuk, hogy az egyetlen zérushely öt tizedes jegy pontossággal:

$$x = 1.23984$$

A Newton1 eljárás

A Newton1 eljárás akkor áll le, ha vagy eléri a kívánt pontosságot, vagy túllépi az előírt lépésszámot.

Az *eljárás paraméterei*: az *f* függvény, a *kezdo* kezdőérték, *epsilon* a pontosság, *lepesszam* a maximális lépésszám.

```
> Newton1:=proc(f,kezdo,epsilon,lepesszam)
   local regi,uj,p;
   global n:
   Digits:=ceil(abs(log[10](epsilon)))+1;
   regi:=evalf(kezdo):n:=1: print(x[0]=regi);
   uj:=regi-f(regi)/D(f)(regi);print(x[1]=uj):
    while(abs(regi-uj))>epsilon and n<lepesszam do
     n:=n+1:
    regi:=uj:
    uj:=regi-f(regi)/D(f)(regi):
      print(x[n]=uj)
   od;print(`Az iterációs lépések száma`=n):
  print(`A  gyök`=uj)
  end:
```

Példa

Keressük meg -mintegy ellenőrzésként is- ezzel az eljárással is az előző szekcióben szereplő függvény zérushelyét! Emlékeztetőül a függvény:

$$\mathrm{f}\colon \mathbf{R}\text{+->}\mathbf{R};\;\; \mathrm{f}(x) = \mathbf{e}^{(-x^2)} - \ln(x)$$

Az előző szekcióban megállapítottuk, hogy az $x = 1$ megfelel kezdőértéknek.

```
> Newton1(f,1,10^(-5),10);
```

$$x_0 = 1.$$

$$x_1 = 1.21194$$

$$x_2 = 1.23940$$

$$x_3 = 1.23984$$

$$x_4 = 1.23984$$

$$\textit{Az iterációs lépések száma} = 4$$

$$\textit{A gyök} = 1.23984$$

```
>
```

A Newtonrajz

Célszerű a numerikus reprezentáció mellett grafikusan is nyomon követni az iterációt. A Newtonrajz eljárás elvégzi az iteráció lépéseit és ábrázolja a közelítő sorozatot. Az *eljárás paraméterei*: az *f* függvény, a *kezdo* kezdőérték, *epsilon* a pontosság, *lepesszam* a maximális lépésszám, *bal* ill *jobb* az ábrázolási intervallum végpontjai.

```
> Newtonrajz:=proc(f,kezdo,epsilon,lepesszam,bal,jobb)
   local regi,uj,p,graf;
   global n:
```

```
 graf:=plot(f(x),x=bal..jobb,color=blue,thickness=2):
 regi:=evalf(kezdo):n:=1:
 uj:=regi-f(regi)/D(f)(regi);
p:=plot([[kezdo,0],[kezdo,f(kezdo)],[uj,0]]):kep||1:=plots[display]([graf,p]):
 while(abs(regi-uj))>epsilon and n<lepesszam do
   n:=n+1:
   regi:=uj:
   uj:=regi-f(regi)/D(f)(regi):
   p:=plot([[regi,0],[regi,f(regi)],[uj,0]]):
   kep||n:=plots[display]([kep||(n-1),p]):
 od;

plots[display]([kep||(1..n)],insequence=true,title=`Az iteráció`);
end:
```

```
>
```

Példa a Newton1 eljárás alkalmazására

Példa

Határozzuk meg az

$$e^{x} = \cos(x) + \sin(2\,x)$$

egyenlet legnagyobb negatív gyökét $\varepsilon = 10^{(-5)}$-nél kisebb hibával!

Megoldás

Az egyenlet gyökeinek meghatározása egyenértékű az $f(x) = e^{x} - \cos(x) - \sin(2\,x)$ függvény zérushelyeinek meghatározásával.

1. Lépés:

Vegyük föl és ábrázoljuk az f függvényt 0 alkalmas intervallumon, hogy a keresett gyököt elkülöníthessük!

```
> f:=x->exp(x)-cos(x)-sin(2*x);
```

$$f := x \rightarrow e^{x} - \cos(x) - \sin(2\,x)$$

```
> plot(f(x),x=-5..0.1,color=[blue,red,green],thickness=2);
```



2. Lépés

Láthatóan a keresett gyök a $[-2, -1]$ intervallumban van. Nézzük meg, hogy ebben az intervallumban teljesülnek-e a megkívánt feltételek, azaz ábrázoljuk az f függvényt és első két deriváltját!

```
> plot([f(x),D(f)(x),(D@@2)(f)(x)],x=-2..-1,color=[blue,red,
  green],thickness=2,legend=["f","f '","f ''"]);
```



1. Legyen f az $[\,a,\,b\,]$ intervallumon kétszer deriválható (akkor folytonos is).

Az f függvény tetszőleges sokszor deriválható függvényekből épül fel, hiszen az x, az $\mathbf{e}^x$ függvény és a szögfüggvények ilyen függvények, és az alapműveletek valamint az összetett függvény képzése megőrzi ezt a tulajdonságot.

2. Ne legyen az f deriváltjának zérushelye, azaz $\frac{d}{dx}\,\mathrm{f}(x) \neq 0$ az [a,b] intervallumon (akkor f szigorúan monoton).

Láthatóan az elsőderivált az intervallum egészében pozivív, tehát az f függvény itt szigorúan monoton növekvő.

3. Ne legyen a második deriváltnak zérushelye, azaz $\frac{d^2}{dx^2}\,\mathrm{f}(x) \neq 0$ az $[a,b]$ intervallumon (akkor f-nek nincs inflexiós pontja).

A második derivált az intervallumban előjelet vált. Emiatt szűkítenünk kell az intervallumot. A $[-2,\,-1.6\,]$ intervallum megfelel, mert tartalmazza a keresett zérushelyet, és ebben az intervallumban már a második derivált is állandó előjelű, ti. pozitív. Készítsük el az f és deriváltjainak ábráját ebben az intervallumban!

```
> plot([f(x),D(f)(x),(D@@2)(f)(x)],x=-2..-1.6,color=[blue,re
  d,green],thickness=2,legend=["f","f '","f ''"]);
>
```



3. Lépés

Kezdőérték választása: A függvényérték és a második derivált előjele az intervallum jobboldali végpontjában egyezik meg, tehát ezt választhatjuk kezdőértékként:

4. Lépés

Alkalmazzuk a **Newton1** eljárást. Ennek paraméterei rendre: az f függvény, a kezdőérték, az ε pontosság és a maximálisan elvégzendő lépésszám (biztosan legyen vége az eljárásnak)

```
> Newton1(f,-1.6,0.00001,10);
```

$$x_0 = -1.6$$

$$x_1 = -1.74407$$

$$x_2 = -1.75113$$

$$x_3 = -1.75116$$

$$x_4 = -1.75116$$

$$\textit{Az iterációs lépések száma} = 4$$

$$\textit{A gyök} = -1.75116$$

A legnagyobb negatív gyök a kívánt pontossággal tehát -1.75116. Ellenőrzésként megvizsgálhatjuk a függvény értékét ezen a helyen.

```
> f(-1.75116);
```

$$0.49805\ 10^{-5}$$

Újabb példák

Példa.

A következő függvény példáján figyeljük meg, hogy ha a feltételek nem teljesülnek, akkor véletlenszerűvé válhat, hogy az eljárás melyik gyököt adja meg.
A függvény legyen az

$$f(x) = x\,(\sin(x) - \cos(2\,x))$$

Megoldás:

```
> f:=x->x*(sin(x)-cos(2*x));
```

$$f := x \to x\,(\sin(x) - \cos(2\,x))$$

```
> plot(f(x),x=-4..5);
```




```
>
```

Legyen a kezdőérték rendre 1.2, 1.64, 1.65, 1.7, 1.85, 2.2. Írjuk rendre ezeket az értékeket az eljárás második paramétereként, s figyeljük meg, hogy az eljárás mindig más és más zérushelyet talál meg!

```
> start:=[1.2, 1.64, 1.65, 1.7, 1.85,  2.2];
```

$$start := [1.2, 1.64, 1.65, 1.7, 1.85, 2.2]$$

```
> epsilon:=0.000001;lépés:=10;
```

$$\varepsilon := 0.1\ 10^{-5}$$

$$lépés := 10$$

```
> Newton1(f,start[1],epsilon,lépés);
```

$$x_0 = 1.2$$

$$x_1 = 0.6622500$$

$$x_2 = 0.5494115$$

$$x_3 = 0.5249146$$

$$x_4 = 0.5236026$$

$$x_5 = 0.5235988$$

$$x_6 = 0.5235986$$

Az iterációs lépések száma = 6

A gyök = 0.5235986

```
> Newtonrajz(f,start[1],epsilon,lépés,-0.1,1.2);
```



```
> Newton1(f,start[2],epsilon,lépés);
```

$$x_0 = 1.64$$

$$x_1 = -0.652640$$

$$x_2 = 3.758764$$

$$x_3 = 4.853074$$

$$x_4 = 4.783167$$

$$x_5 = 4.747965$$

$$x_6 = 4.730235$$

$$x_7 = 4.721327$$

$$x_8 = 4.716859$$

$$x_9 = 4.714623$$

$$x_{10} = 4.713504$$

Az iterációs lépések száma = 10

A gyök = 4.713504

```
> Newtonrajz(f,start[2],epsilon,lépés,-3,5.3);
```




```
> Newton1(f,start[3],epsilon,lépés);
```

$$x_0 = 1.65$$

$$x_1 = -0.805808$$

$$x_2 = -2.281456$$

$$x_3 = -1.898985$$

$$x_4 = -1.741332$$

$$x_5 = -1.659091$$

$$x_6 = -1.615951$$

$$x_7 = -1.593666$$

$$x_8 = -1.582310$$

$$x_9 = -1.576575$$

$$x_{10} = -1.573690$$

$$\textit{Az iterációs lépések száma} = 10$$

$$\textit{A gyök} = -1.573690$$

```
> Newtonrajz(f,start[3],epsilon,lépés,-3,1.7);
```




```
>
```

A további kezdőértékkel végezze el a felhasználó a kísérletezést! Ehhez csupán az előzőkben látott lépéssorozatot kell ismételgetni a megfelelő kezdőértékkel.

Újabb példák

1. Példa

1. Határozza meg az $x^3 + 5\,x - 3 = 0$ egyenlet pozitív gyökét!

```
> f:=x->x^3+5*x-3;
```

$$f := x \rightarrow x^3 + 5\,x - 3$$

```
> D(f)(x);
```

$$3\,x^2 + 5$$

```
> (D@@2)(f)(x);
```

$$6\,x$$

```
> plot(f(x),x=0..2);
```




```
> newton(f,2,5);
```

$$\begin{bmatrix} n & x_n & \mathrm{f}(x_n) & \mathrm{D}(f)(x_n) & x_n - \dfrac{\mathrm{f}(x_n)}{\mathrm{D}(f)(x_n)} \\ 1 & 2. & 15. & 17. & 1.117647059 \\ 2 & 1.117647059 & 3.984327296 & 8.747404844 & 0.6621602755 \\ 3 & 0.6621602755 & 0.601129676 & 6.315368691 & 0.5669750685 \\ 4 & 0.5669750685 & 0.017135560 & 5.964382185 & 0.5641020869 \\ 5 & 0.5641020869 & 0.000014016 & 5.954633493 & 0.5640997331 \end{bmatrix}$$

```
> Newton1(f,1,10^(-5),10);
```

$$x_0 = 1.$$

$$x_1 = 0.625000$$

$$x_2 = 0.565190$$

$$x_3 = 0.564101$$

$$x_4 = 0.564101$$

Az iterációs lépések száma = 4

A gyök = 0.564101

```
> Newtonrajz(f,3,10^(-3),5,0,3);
```



2. Példa

Határozza meg a $cos(x) = x$ egyenlet gyökét!

```
> plot([cos(x),x],x=0..1,color=[blue,red]);
```




```
> f:=x->cos(x)-x;
```

$$f := x \rightarrow \cos(x) - x$$

```
> D(f)(x);
```

$$-\sin(x) - 1$$

```
> (D@@2)(f)(x);
```

$$-\cos(x)$$

```
> plot([f(x),D(f)(x),(D@@2)(f)(x)],x=0..2,color=[blue,red,gr
een]);
```




```
> newton(f,1,5);
```

$$\begin{bmatrix} n & x_n & f(x_n) & D(f)(x_n) & x_n - \dfrac{f(x_n)}{D(f)(x_n)} \\ 1 & 1. & -0.4596976941 & -1.841470985 & 0.7503638679 \\ 2 & 0.7503638679 & -0.0189230739 & -1.681904953 & 0.7391128909 \\ 3 & 0.7391128909 & -0.0000464559 & -1.673632544 & 0.7390851334 \\ 4 & 0.7390851334 & -0.3\ 10^{-9} & -1.673612029 & 0.7390851332 \\ 5 & 0.7390851332 & 0. & -1.673612029 & 0.7390851332 \end{bmatrix}$$

```
> Newton1(f,1,10^(-5),6);
```

$$x_0 = 1.$$
$$x_1 = 0.750364$$
$$x_2 = 0.739113$$
$$x_3 = 0.739085$$
$$x_4 = 0.739085$$
$$\textit{Az iterációs lépések száma} = 4$$
$$\textit{A gyök} = 0.739085$$

```
>
```

3. Példa

Adja meg az $x^4 - 5x^2 + 2x - 5 = 0$ egyenlet 2 és 3 közé eső gyökét!

```
> f:=x->x^4-5*x^2+2*x-5;
```

$$f := x \rightarrow x^4 - 5\,x^2 + 2\,x - 5$$

```
> f(2);
```

$$-5$$

```
> f(3);
```

$$37$$

```
> D(f)(x);
```

$$4\,x^3 - 10\,x + 2$$

```
> (D@@2)(f)(x);
```

$$12\,x^2 - 10$$

```
> plot([f(x),D(f)(x),(D@@2)(f)(x)],x=2..3,color=[blue,red,green]);
```




```
> newton(f,3,6);
```

n	x_n	$f(x_n)$	$D(f)(x_n)$	$x_n - \frac{f(x_n)}{D(f)(x_n)}$
1	3.	37.	80.	2.537500000
2	2.537500000	9.33998245	41.97989844	2.315012981
3	2.315012981	1.555537262	28.47712851	2.260388890
4	2.260388890	0.079526610	25.59265474	2.257281490
5	2.257281490	0.000247460	25.43346870	2.257271760
6	2.257271760	$-0.10\ 10^{-7}$	25.43297108	2.257271760

>

Kérdések, feladatok

Ellenőrző kérdések

1. Milyen feltételek teljesülése esetén alkalmazhatjuk a Newton-féle érintőmódszert?
2. Mit jelent a gyökök elkülönítése?

3.Mit biztosít az a feltétel, amely szerint $\frac{d}{dx} f(x) \neq 0$ az [a,b] intervallumon?

4. Miért van szükség a $\frac{d^2}{dx^2} f(x) \neq 0\ x \in [a, b]$ feltételre?

5. Hogyan kell a kezdőértéket megválasztani?

Gyakorló feladatok

Oldja meg a következő feladatokat a Newton-féle érintő módszerrel és szemléltesse a munkalapban látott módon az iterációt!

1. Határozza meg az $x^3 + 5x - 3 = 0$ egyenlet pozitív gyökét!
2. Adja meg az $x^3 - 3x + 1 = 0$ egyenlet legnagyobb gyökét!
3. Határozza meg az $x^4 + 2x^3 - 5x^2 + 1 = 0$ egyenlet 1 és 2 közé eső gyökét!
4. Adja meg az $x^4 - 5x^2 + 2x - 5 = 0$ egyenlet 2 és 3 közé eső gyökét!
5. Határozza meg a $cos(x) = x$ egyenlet gyökét!
6. Adja meg a $\cos(x) + x = 2$ egyenlet gyökét!
7. Számítsa ki az $\mathbf{e}^x - \sin(x) = 0$ egyenlet legnagyobb gyökét!
8. Számítsa ki az alábbi egyenletek összes valós gyökét!

a) $x^4 - x - 2 = 0$; b) $x^5 - 2x^2 + 4 = 0$;

c) $2x^3 + \mathbf{e}^x = 0$; d) $2\,\mathbf{e}^x + x - 1 = 0$;

e) $2x - 5 - \sin(x) = 0$; e) $x^2 - 2 - \mathbf{e}^{(-2x)} = 0$.